

NAG Library Routine Document

F03BHF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F03BHF computes the determinant of a n by n symmetric positive definite banded matrix A that has been stored in band-symmetric storage. F07HDF (DPBTRF) must be called first to supply the Cholesky factorized form. The storage (upper or lower triangular) used by F07HDF (DPBTRF) is relevant as this determines which elements of the stored factorized form are referenced.

2 Specification

```
SUBROUTINE F03BHF (UPLO, N, KD, AB, LDAB, D, ID, IFAIL)
INTEGER          N, KD, LDAB, ID, IFAIL
REAL (KIND=nag_wp) AB(LDAB,*), D
CHARACTER(1)    UPLO
```

3 Description

The determinant of A is calculated using the Cholesky factorization $A = U^T U$, where U is an upper triangular band matrix, or $A = LL^T$, where L is a lower triangular band matrix. The determinant of A is the product of the squares of the diagonal elements of U or L .

4 References

Wilkinson J H and Reinsch C (1971) *Handbook for Automatic Computation II, Linear Algebra* Springer-Verlag

5 Parameters

- 1: UPLO – CHARACTER(1) *Input*
On entry: indicates whether the upper or lower triangular part of A was stored and how it was factorized. This should not be altered following a call to F07HDF (DPBTRF).
 UPLO = 'U'
 The upper triangular part of A was originally stored and A was factorized as $U^T U$ where U is upper triangular.
 UPLO = 'L'
 The lower triangular part of A was originally stored and A was factorized as LL^T where L is lower triangular.
Constraint: UPLO = 'U' or 'L'.
- 2: N – INTEGER *Input*
On entry: n , the order of the matrix A .
Constraint: $N > 0$.

- 3: KD – INTEGER *Input*
On entry: k_d , the number of superdiagonals or subdiagonals of the matrix A .
Constraint: $KD \geq 0$.
- 4: AB(LDAB,*) – REAL (KIND=nag_wp) array *Input*
Note: the second dimension of the array AB must be at least $\max(1, N)$.
On entry: the Cholesky factor of A , as returned by F07HDF (DPBTRF).
- 5: LDAB – INTEGER *Input*
On entry: the first dimension of the array AB as declared in the subprogram from which F03BHF is called.
Constraint: $LDAB \geq KD + 1$.
- 6: D – REAL (KIND=nag_wp) *Output*
 7: ID – INTEGER *Output*
On exit: the determinant of A is given by $D \times 2.0^{ID}$. It is given in this form to avoid overflow or underflow.
- 8: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**
On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, UPLO \neq 'L' or 'U'.

IFAIL = 2

On entry, $N < 1$.

IFAIL = 3

On entry, $KD < 0$.

IFAIL = 5

On entry, $LDAB < KD + 1$.

IFAIL = 6

The matrix A is not positive definite.

7 Accuracy

The accuracy of the determinant depends on the conditioning of the original matrix. For a detailed error analysis see page 54 of Wilkinson and Reinsch (1971).

8 Further Comments

The time taken by F03BHF is approximately proportional to n .

This routine should only be used when $m \ll n$ since as m approaches n , it becomes less efficient to take advantage of the band form.

9 Example

This example calculates the determinant of the real symmetric positive definite band matrix

$$\begin{pmatrix} 5 & -4 & 1 & & & & \\ -4 & 6 & -4 & 1 & & & \\ 1 & -4 & 6 & -4 & 1 & & \\ & 1 & -4 & 6 & -4 & 1 & \\ & & 1 & -4 & 6 & -4 & 1 \\ & & & 1 & -4 & 6 & -4 \\ & & & & 1 & -4 & 5 \end{pmatrix}.$$

9.1 Program Text

```

Program f03bhfe

!      F03BHF Example Program Text

!      Mark 24 Release. NAG Copyright 2012.

!      .. Use Statements ..
Use nag_library, Only: dpbtrf, f03bhf, nag_wp, x04cef
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
Real (Kind=nag_wp)         :: d
Integer                     :: i, id, ifail, info, j, kd, kl, ku,    &
                             ldab, n
Character (1)               :: uplo
!      .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: ab(:, :)
!      .. Intrinsic Procedures ..
Intrinsic                   :: index, max, min
!      .. Executable Statements ..
Write (nout,*) 'F03BHF Example Program Results'

!      Skip heading in data file
Read (nin,*)

Read (nin,*) uplo

Read (nin,*) n, kd
ldab = kd + 1
Allocate (ab(ldab,n))

If (index('Ll',uplo)<=0) Then
!      Read in upper triangular banded matrix
ku = kd
kl = 0
Do i = 1, n
    Read (nin,*) (ab(kd+1+i-j,j),j=i,min(i+kd,n))
End Do

```

```

      Else
!      Read in lower triangular banded matrix
      ku = 0
      kl = kd
      Do i = 1, n
        Read (nin,*) (ab(1+i-j,j),j=max(1,i-kd),i)
      End Do
    End If
!      Factorize A
!      The NAG name equivalent of dpbtrf is f07hdf
    Call dpbtrf(uplo,n,kd,ab,ldab,info)

    If (info==0) Then
      Write (nout,*)
      Flush (nout)
      ifail = 0
      Call x04cef(n,n,kl,ku,ab,ldab,'Array AB after factorization',ifail)

      ifail = 0
      Call f03bhf(uplo,n,kd,ab,ldab,d,id,ifail)

      Write (nout,*)
      Write (nout,99999) d, id
      Write (nout,*)
      Write (nout,99998) d*2.0E0_nag_wp**id
    Else
      Write (nout,99997) info
    End If

99999 Format (1X,'D = ',F13.5,' ID = ',I0)
99998 Format (1X,'Value of determinant = ',E13.5)
99997 Format (' ** Factorization routine return error flag info = ',I0,'.')
    End Program f03bhfe

```

9.2 Program Data

F03BHF Example Program Data

```

L          : UPLO
7 2       : N, KD
5
-4        6
1   -4    6
      1   -4  6
          1  -4  6
              1 -4  6
                  1 -4 5 : AB

```

9.3 Program Results

F03BHF Example Program Results

Array AB after factorization

	1	2	3	4	5	6	7
1	2.2361						
2	-1.7889	1.6733					
3	0.4472	-1.9124	1.4639				
4		0.5976	-1.9518	1.3540			
5			0.6831	-1.9695	1.2863		
6				0.7385	-1.9789	1.2403	
7					0.7774	-1.9846	0.6761

D = 0.25000 ID = 8

Value of determinant = 0.64000E+02